
Programkonstruktion
och datastrukturer

Moment 3
Rekursion 
och listor

PKD 2010/11 moment 3 Sida 1 Uppdaterad 2011-02-08



Summera godtyckligt antal tal
sumUpTo n skall beräkna summan av talen från 0 till n.

fun sumUpTo 0 = 0
  | sumUpTo 1 = 0+1
  | sumUpTo 2 = 0+1+2
  | sumUpTo 3 = 0+1+2+3
  | sumUpTo 4 = 0+1+2+3+4
  | sumUpTo 5 = 0+1+2+3+4+5
  | sumUpTo 6 = 0+1+2+3+4+5+6
  | sumUpTo 7 = 0+1+2+3+4+5+6+7;

n>7, då?? Detta är uppenbarligen ingen framkomlig väg...

Vi behöver en metod att upprepa additionen så att godtyckligt 
många tal kan adderas.

Analys av summeringsproblemet
• För att summera talen från 0 till n kan jag lösa det enklare 
  problemet att summera från 0 till n-1 och sedan lägga till n!
• Om n=0 kan jag direkt säga att summan är 0.
Exempel: summera talen från 0 till 4.
• Ett enklare problem är att summera från 0 till 3.
• Summan av det enklare problemet blir 6?
• Lägg till 4. Nu har vi 6+4=10 vilket är summan av talen från 0 till 4.
Hur fick jag tag på summan av talen från 0 till 3? På samma sätt!
• Ett enklare problem är att summera från 0 till 2.
• Summan av det enklare problemet blir 3?
• Lägg till 3. Nu har vi 3+3=6 vilket är summan av talen från 0 till 3.
Etcetera…

Fullständigt resonemang för summeringen
Summera talen från 0 till 4.
• Ett enklare problem är att summera från 0 till 3.
•• Ett enklare problem är att summera från 0 till 2.
••• Ett enklare problem är att summera från 0 till 1.
•••• Ett enklare problem är att summera från 0 till 0.
••••• Summan är 0!
•••• Summan är 0+1 = 1.
••• Summan är 1+2 = 3.
•• Summan är 3+3 = 6.
• Summan är 6+4=10.
Denna metod kallas rekursion.
(Från latinets recurso – att återvända.)

Rekursion i ML
fun sumUpTo(0) = 0
  | sumUpTo(n) = sumUpTo(n-1) + n

Rekursivt anropBasfall

sumUpTo 4
––> sumUpTo(4-1)+4
––> sumUpTo 3 +4
––> (sumUpTo(3-1)+3)+4
––> (sumUpTo 2 +3)+4
––> ((sumUpTo(2-1)+2)+3)+4
––> ((sumUpTo 1 +2)+3)+4
––> (((sumUpTo(1-1)+1)+2)+3)+4
––> (((sumUpTo 0 +1)+2)+3)+4
––> (((0+1)+2)+3)+4
––> …… 
––> 10 
Rekursion är en programmeringsteknik.

Hur skall man förstå detta?

fun sumUpTo 0 = 0
  | sumUpTo n =
      sumUpTo(n-1) + n

Rekursion är en variant av principen att dela upp problemet i 
enklare delar och sedan sätta ihop resultatet av delarna.
En rekursiv funktion anropar sig själv, men problemet är ändå 
enklare eftersom argumenten är ”mindre” (precis vad detta betyder 
återkommer vi till). När det rekursiva anropet är klart sätter man ihop 
resultatet (additionen i exemplet).
Varje rekursivt anrop är som en ”rysk docka”.

Bindningar i rekursion
fun sumUpTo 0 = 0
  | sumUpTo n = sumUpTo(n-1) + n
sumUpTo 2
––> sumUpTo(n-1)+n              [n-->2]
––> sumUpTo(2-1)+n              [n-->2]
––> sumUpTo 1 +n                [n-->2]
––> (sumUpTo(n-1)+n [n-->1])+n  [n-->2]
––> (sumUpTo(1-1)+n [n-->1])+n  [n-->2]
––> (sumUpTo 0 +n   [n-->1])+n  [n-->2]
––> (0+n [n-->1])+n             [n-->2]
––> (0+1 [n-->1])+n             [n-->2]
––> (1 [n-->1])+n               [n-->2]
––> 1+n                         [n-->2]
––> 1+2                         [n-->2]
––> 3                           [n-->2]
––> 3

Fakultetsfunktionen
Standardexempel på rekursion i programmering är 
fakultetsfunktionen n!.
n! = 1*2*…*n, för heltal n>0. Funktionen fact beräknar n!
fun fact 0 = 1
  | fact n = n*fact(n-1)

fact växer fort – fact 13 ger overflow på institutionens maskiner.
fact är i princip samma funktion som sumUpTo, men med 
multiplikation i stället för addition.

 

PKD 2010/11 moment 3 Sida 8 Uppdaterad 2011-02-08



Euklides algoritm

Låt r vara resten när m delas med n.

Är n=0? Stopp.

Start.

Ja.

Nej.

Låt m vara n.

Låt n vara r.
För att beräkna gcd för m och n, lös det enklare problemet att 
beräkna gcd för n och resten vid division av m med n!

gcd i ML
gcd(m,n) beräknar största gemensamma delaren till m och n, för  
heltal m, n större än 0.

fun gcd(m,0) = m
  | gcd(m,n) = gcd(n, m mod n)

gcd(126,168)
––> gcd(168,126 mod 168)
––> gcd(168,126)
––> gcd(126,168 mod 126)
––> gcd(126,42)
––> gcd(42,126 mod 42)
––> gcd(42,0)
––> 42 

Terminering
Hur vet man att rekursionen någonsin avslutas (terminerar)?
Vad innebär det att den inte gör det?
Inte självklart att rekursionen terminerar... felaktiga argument:
sumUpTo ~1
––> sumUpTo(~1-1) + ~1
––> sumUpTo(~2) + ~1
––> (sumUpTo(~2-1) + ~2) + ~1
––> (sumUpTo(~3) + ~2) + ~1
––> ..... 
…eller en liten felskrivning: 
fun sumUpTo(0) = 0
  | sumUpTo(n) = sumUpTo(n+1) + n

Rekursionen måste vara enklare
Varje rekursivt anrop måste lösa ett enklare fall än det föregående. 
Ett tillräckligt enkelt fall måste kunna lösas utan rekursion.
Varje rekursivt anrop av sumUpTo n  har ett mindre värde av n än 
det föregående. Fallet när n=0 (basfallet) hanteras utan 
rekursion. Samtidigt kan inte n bli mindre än 0.
Om det rekursiva anropet alltid blir enklare och det finns ett enklaste 
fall som löses utan rekursion så är terminering garanterad.

Vi kommer att längre fram göra mera precist vad ”enklare” betyder.

gcd terminerar
fun gcd(m,0) = m
  | gcd(m,n) = gcd(n, m mod n)

Det rekursiva anropet är enklare eftersom n minskar.

• m mod n är alltid (definitionsmässigt) mindre än n.
• n kan inte vara mindre än 0 eftersom vi förutsatte att m och n
  från början är större än 0 och m mod n inte kan bli mindre
  än 0).
• fallet n=0 är det ”enklaste” fallet som hanteras utan rekursion.

Exempel: räkna uppåt till basfallet
sumRange(i,n)beräknar summan av heltalen från i till n.
fun sumRange(i,n) = if i>n then
                      0
                    else
                      sumRange(i+1,n)+i;

Det rekursiva anropet blir enklare för att antalet tal som återstår att 
addera blir mindre.
Alternativ definition av sumUpTo med hjälp av sumRange:
fun sumUpTo n = sumRange(0,n);

Varför har jag inte skrivit sumRange så här?
fun sumRange(i,n) = if i=n then
                      i
                    else
                      sumRange(i+1,n)+i;
PKD 2010/11 moment 3 Sida 14 Uppdaterad 2011-02-08



Tolkning av konstiga ”summor”
Summerar man två delar av en talserie och adderar delsummorna 
får man summan av hela talserien.

1, 2, 3, 4, 5, 6

1+2+3+4+5+6=21

1+2+3=6 4+5+6=15 6+15=21      

1, 2, 3, 4, 5, 6

1+2+3+4+5+6=21

1=1 2+3+4+5+6=20 1+20=21
Låt nu en delserie innehålla ett tal. Regeln gäller fortfarande om 
man antar att summan av ett tal är talet självt.
Låt en delserie vara tom – dvs innehålla noll tal. Regeln 
gäller fortfarande om man antar att summan av noll tal är 0.
En talserie bestäms av sitt första och sista tal. En serie med noll tal 
får ett ”sista” tal (n) som är ett mindre än det ”första” talet (i).

Undvik onödiga begränsningar
sumRange(i,n) summerar tal från i till n.
Som vi såg på förra bilden så kan man tala om en summa av ”noll” 
tal om n=i-1. Eftersom sumRange kodades
fun sumRange(i,n) = if i>n then
                      0
                    else
                      sumRange(i+1,n)+i;

…så ger den rätt resultat för en summa av ”noll” tal.
Man bör försöka hitta en naturlig tolkning av funktionens uppgift som 
är så generell som möjligt utan att krångla till det. Då får man ett 
program som är flexibelt och kan användas i många situationer.

Vanliga fel i algoritmbeskrivningar
Rekursion handlar om upprepning. Beskrivningar av algoritmer med 
upprepning måste man göras extra noga – speciellt med syftningar.
Felaktig algoritmbeskrivning för funktionen sumUpTo:
   1. Är talet 0 så är det summan.
   2. Räkna ut summan för föregående tal.
   3. Addera summan till talet.
• Vad syftar ”talet” på i steg 1 och 3? 
• Vad syftar ”det” på i steg 1? Vad händer om ”talet” inte är 0?
• Vad är ”föregående tal” i steg 2? Antagligen ”talet”-1, men kan 
   betyda att gå till föregående tal i en följd av olika tal.
• Hur förstår man att steg 2 innebär en upprepning (rekursion)?
• Var finns resultatet när algoritmen avslutas?

Bättre algoritmbeskrivning
Felaktig algoritmbeskrivning för funktionen sumUpTo.
   1. Är talet 0 så är det summan.
   2. Räkna ut summan för föregående tal.
   3. Addera summan till talet.
Riktig algoritmbeskrivning för funktionen sumUpTo.
   Algoritm för att beräkna summan av heltalen från 0 till n.
   1. Är n=0? I så fall stoppa, summan är 0.
   2. Använd algoritmen rekursivt för att summera talen från 0 till n-1.
   3. Addera talet n till resultatet av steg 2.
   4. Stoppa. summan är resultatet av steg 3.

Rekursion över strängar
countChar(s,c) räknar antalet förekomster av c i s. Till exempel:
   countChar("Hej, du glade",#"d")=2
   countChar("Hej, du glade",#"q")=0

fun countChar("",_) = 0
  | countChar(s,c) =
      countChar(String.substring(s,1,size s -1),c)
      + (if String.sub(s,0)=c then
           1
         else
           0);

Det rekursiva anropet är enklare eftersom det räknar bland färre 
tecken. Första argumentet är ett tecken kortare.

En annan ansats till teckenräkning
Vid rekursion över strängar är rekursionsargumentet en delsträng.
Delsträngen behöver inte konstrueras uttryckligen – istället kan den 
vara underförstådd (implicit) genom att använda en position i 
strängen som extra argument och göra rekursion över positionen.
Strängargumentet s ersätts av två argument s och pos.
s är den ursprungliga strängen, pos är positionen.
När man använder s får man skriva
    String.sub(s,pos) istället för String.sub(s,0)
Det rekursiva anropet har
    s och pos+1 istället för String.substring(s,1,size s-1)
Denna teknik är viktig om det är komplicerat att konstruera uttrycket 
i det rekursiva anropet.

Rekursion över position i en sträng
countChar2'(s,c,pos) är antalet förekomster av c i s från och 
med positionen pos.
fun countChar2'(s,c,pos) =
      if pos >= size s then
        0
      else
        countChar2'(s,c,pos+1)
         + (if String.sub(s,pos)=c then 1 else 0);

Det rekursiva anropet är enklare för att det finns färre tecken kvar  
att titta på mellan pos och slutet av strängen.
countChar2(s,c) är antalet förekomster av c i s, men 
beräkningen sker på ett annat sätt än i countChar.
fun countChar2(s,c) = countChar2'(s,c,0);

Räkna tecken baklänges
countChar3'(s,c,pos) är antalet förekomster av c i s från och 
med strängens början och pos tecken framåt.
Basfallet kan hanteras med matchning.
fun countChar3'(s,c,0) = 0
  | countChar3'(s,c,pos) =
      countChar3'(s,c,pos-1)
      + (if String.sub(s,pos-1)=c then 1 else 0);

Det rekursiva anropet är enklare för att det finns färre tecken kvar  
att titta på mellan början av strängen och pos.
countChar3(s,c) är antalet förekomster av c i s, men 
beräkningen sker på ett annat sätt än i countChar.
fun countChar3(s,c) = countChar3'(s,c,size s);

Tornen i Hanoi

A B C
Problem: Flytta alla skivor från A till C.
Regler: En skiva i taget får flyttas
            Endast den översta skivan får flyttas.
            En skiva får bara ligga ovanpå större skivor.

Uppgift: Skriv ett program för att beräkna hur flyttningen skall gå till 
(givet ett visst antal brickor).

(Ett exempel på ett planeringsproblem)

Lösning av Hanoi-problemet
Vi skall flytta n brickor.
Antag att vi vet hur man skall flytta n-1 brickor.
• Flytta n-1 brickor från A till B (via C).
• Flytta den sista brickan från A till C.
• Flytta n-1 brickor från B till C (via A).
Klart!!

När man löser delproblemen är det inte säkert att ”till” eller ”via”-
pinnarna är tomma, men i så fall ligger det större brickor på dem.

Denna lösning ger direkt ett naturligt program i ML.
Observera att lösningen kräver två rekursiva anrop (multipel 
rekursion).

Hanoi-problemet i ML
hanoi(n,from,via,to) beräknar en sträng som beskriver hur 
man skall flytta n brickor i Hanoi-spelet från pinne from till pinne to 
med hjälp av pinne via.
T.ex. hanoi(2,"A","B","C") = "A->B A->C B->C "
fun hanoi(0,_,_,_) = ""
  | hanoi(n,from,via,to) =
      hanoi(n-1,from,to,via) ^
      from ^ "->" ^ to ^ " " ^
      hanoi(n-1,via,from,to);

Ömsesidig rekursion
Använd rekursion för att visa om ett naturligt tal är udda eller jämnt.
even n är true om n är jämnt, false annars.
odd n är true om n är udda, false annars.

Utnyttja att om n är jämnt så är n-1 udda och tvärtom!
fun even 0 = true
  | even n = odd (n-1);
fun odd 0 = false
  | odd n = even (n-1);

Detta kallas ömsesidig rekursion och leder till ett problem eftersom 
båda funktionerna använder den andra – båda måste alltså vara 
definierade före den andra?!?

Ömsesidigt rekursion – problem
Programmet ovan fungerar inte direkt i ML – odd används (i even) 
innan den är definierad. Ömsesidigt rekursiva funktioner måste 
definieras samtidigt med hjälp av and-konstruktionen:
- fun even 0 = true 
=   | even n = odd (n-1)
= and odd 0 = false
=   | odd n = even (n-1);
val even = fn : int -> bool
val odd = fn : int -> bool
- even 5;
val it = false : bool
- even 4;
val it = true : bool

Rekursion är inte alltid direkt tillämplig
Problem: Avgör om n är ett primtal!
Problemuppdelning? Blir det enklare att avgöra om n är ett primtal 
om man vet att n-1 (eller något annat tal mindre än n) är ett primtal?
n är tydligen inte rätt värde att göra rekursion över, men vad är rätt i 
så fall?

Generalisering
Ibland kan man inte lösa ett problem direkt, utan man måste lösa ett 
mera allmänt problem först.
Problem: avgör om n inte är delbart med något heltal i talintervallet 
low…high.
Detta är enkelt att lösa rekursivt. Välj ut ett av talen i intervallets 
gränser (low eller high). Gör sedan rekursion över ett intervall där 
man tagit bort det tal man just undersökt. Intervallet är kortare och 
problemet är därför enklare.
n är ett primtal omm (om och endast om) n inte är delbart med något 
tal utom 1 och sig själv – alltså från 2 till n-1. Vi kan använda det 
mera generella delbarhetsproblemet för att lösa primtalsproblemet!

Programmet för primtalsbestämning
nodivisors(n,low,high) är true om n inte är delbart med 
något tal i intervallet från low till high, false annars.
fun nodivisors(n,low,high) =
      low > high orelse
       (n mod low) <> 0 andalso
        nodivisors(n,low+1,high);

prime n är sant omm n är ett primtal.
fun prime n = nodivisors(n,2,n-1);

Notera basfallet i nodivisors! Ett tomt intervall har man om high 
är low-1. nodivisors är då trivialt sant – n är delbart med alla tal i 
intervallet eftersom intervallet är tomt. (Jämför diskussionen om 
summering av tal i intervall.)

Lokala deklarationer
Man kan låta deklarationer gälla endast inuti ett uttryck.
   let
     deklarationer...
   in
     uttryck
   end
är självt ett uttryck där räckvidden för de angivna deklarationerna 
endast går till slutet av let-uttrycket.
   let
     val x = 3
     val y = 2
   in
     x+y
   end

...beräknas till 5. Definitionerna av x och y är lokala i let-uttrycket.

Minns ni countChar2?
countChar2'(s,c,pos) är antalet förekomster av c i s från och 
med positionen pos.
fun countChar2'(s,c,pos) =
      if pos >= size s then
        0
      else
        countChar2'(s,c,pos+1)
         + (if String.sub(s,pos)=c then 1 else 0);

Det rekursiva anropet är enklare för att det finns färre tecken kvar  
att titta på mellan pos och slutet av strängen.
countChar2(s,c) är antalet förekomster av c i s, men 
beräkningen sker på ett annat sätt än i countChar.
fun countChar2(s,c) = countChar2'(s,c,0);

En användning av lokala deklarationer
fun countChar2(s,c) = 
  let
    val ss = size s;
    fun countChar2' pos =
      if pos >= ss then
        0
      else if String.sub(s,pos)=c then
        countChar2'(pos+1)+1
      else
        countChar2'(pos+1)
  in  countChar2' 0
  end
• size s räknas bara ut en gång.
• countChar2' kan inte användas utanför countChar2
• s, c och ss behöver inte ges som argument till countChar2'
  (de är fria i countChar2').

Matchning i deklarationer
I en val-deklaration är identifierarnamnet egentligen ett mönster!
- val (x,y) = (1,2);
val x = 1 : int
val y = 2 : int
I detta fall måste uttrycket matcha mönstret, annars exekveringsfel!
I en lokal deklaration kan matchning användas för att ta fram 
delarna av ett sammansatt funktionsvärde:
fun qaddSimp(x,y) =
  let
    val (p,q) = qadd(x,y);
    val gcdpq = gcd(p,q)
  in
    (p div gcdpq,q div gcdpq)
  end
(qadd var exempelfunktionen som adderade rationella tal.)

Funktionsdefinitioner är syntaktiskt socker!
Funktionsdefinitioner är funktionsabstraktion+definitionsabstraktion.
Detta kan man göra tydligt genom att göra funktionsabstraktionen 
med fn och definitionsabstraktionen med val!
  val rec addUpTo = fn 0 => 0
                     | n => addUpTo(n-1)+n;

Detta är precis detsamma som:
  fun addUpTo 0 = 0
    | addUpTo n = addUpTo(n-1)+n;

nyckelordet rec tillåter att man använder det bundna namnet 
(addUpTo) i sin egen definition – normalt är det inte tillåtet eftersom 
namn måste bindas innan de används.
(Även nyckelordet rec är syntaktiskt socker, men att visa hur man 
kan ersätta det med annan ML-kod är mycket komplicerat.)

Lokala deklarationer är syntaktiskt socker!
   let
     val v1 = uttryck1
   in
     uttryck
   end

Är exakt samma sak som
(fn v1 => uttryck) uttryck1

Program- och datastrukturer

Programstruktur Datastruktur

Sammansättning: uttryck som argument tupler
Alternativ: if/case/matchning ?? (inte ännu)
Upprepning: rekursion listor (strängar)

                                                           aktuellt ämne

Listor
En lista är en ordnad uppsättning (sekvens, följd) av element, som 
kan ha godtycklig typ.

• [18, 12, ~5, 12, 10] : int list
element

• [2.0, 5.3, 3.7, ~1E5] : real list
• ["Lars", "Henrik", "PK"] : string list
• [(1,"A"), (2,"B")] : (int*string) list
• [Math.sin, fn x=>x+1.0] : (real->real) list
• [[1], [2, 3]] : int list list

Karakteristiskt för listor:
• Alla element i en viss lista måste ha samma typ.
• En lista kan innehålla godtyckligt många element (eller inga alls).
• Samma element kan finnas flera gånger i samma lista.

Praktisk användning av listor
En förteckning (böcker i ett bibliotek):
["Introduction to Programming Using Standard ML",
 "Mathematical Logic", "Software Engineering",
 "Applications of Formal Methods"] : string list

En tabell (telefonanknytningar på inst. för IT):
[("Lars-Henrik Eriksson", 1057),
 ("Jesper Bengtson", 3156),
 ("Mattias Wiggberg", 3176),
 ("Aletta Nylén", 7595)] : (string*int) list

En kalender:
[((11,5),[(10,"PKD"),(13,"PKD"),
          (15,"Möte med Arne")]),
 ((11,6),[(10,"LG-möte"),(13,"PKD")])
] : ((int*int)*(int*string) list) list

Listuttryck
Precis som tupler kan listor konstrueras av uttryck som beräknar ett 
visst värde.

• [18, 3+9, 5-7, size("abc"), 10]
     ––> [18, 12, ~2, 3, 10]
• ["A"^"B", Int.toString(5)]
   ––> ["AB", "5"]
• [(1+1,2), qplus((1,2),(1,3))]
   ––> [(2,2), (5,6)]

En tom lista: []

Struktur hos listor
head   tail

cons-cell                   
Listor byggs upp som kedjor. Ex. [18, 12, ~5, 12, 10]

nil
18 12 ~5

12 10
Man kommer bara åt element genom att gå igenom länkarna. Tids-
åtgången (antal steg) är proportionell mot elementets plats i listan.
Pilarna (pekarna) kan inte ändras. Listor måste byggas bakifrån!
I ML (och t.ex. Java) är pekarna implicita – de är ett sätt att beskriva 
datastrukturer men är inte värden i språket. I en del andra språk 
(t.ex. C) är de explicita, dvs. separata värden med egen datatyp.

Grundläggande operationer på listor.
hd (head) ger första elementet i en lista..
  hd [1,2,3] ––> 1                            (Resultatet blir inte [1])
  hd [] ger exekveringsfel (Empty).
tl (tail) ger en lista utom första elementet.
  tl [1,2,3] ––> [2,3]
  tl ["Hej"] ––> []
  tl [] ger exekveringsfel (Empty).
null talar om ifall en lista är tom.
  null [1,2,3] ––> false
  null (tl ["Hej"]) ––> true
:: (cons) skapar en ny listcell av ett element och en lista.
  1::[2,3] ––> [1,2,3]         (:: är en infix operator)
[1,2,3] är egentligen 1::(2::(3::[]))), vilket även kan 
skrivas 1::2::3::[] (:: är högerassociativ).

Polymorfa typer
hd [1,2,3] ––> 1, hd ["Kalle","Anka"] ––> "Kalle"
Vilken typ har hd? int list->int? string list->string?
hd har den polymorfa (mångformiga) typen

'a list -> 'a
Där 'a är en typvariabel. Alla typvariabler i ML har namn som börjar 
med apostrof.
hd kan användas i alla sammanhang där det behövs en typ som 
man kan få genom att ersätta 'a med en annan typ, t.ex. int eller 
string. (Detta kallas en instans av 'a list -> 'a).
En icke polymorf typ kallas för en monomorf (enkelformig) typ.
Polymorfi är inte samma sak som överlagring (t.ex. hos +, <).

Typen av tomma listan
Vilken typ har den tomma listan, []?

[] kan vara en tom lista av heltal, en tom lista av strängar etc....

tl [1] ––> [], tl ["Kalle"] ––> []

[] måste alltså höra till både typen int list och string list 
(och alla andra sorters listor).

[] är ett värde som har den polymorfa typen 'a list.

Polymorfi med likhetstyper
Även funktionerna för (o)likhetsjämförelser är polymorfa.
= : ''a*''a -> bool
<>: ''a*''a -> bool

Dubbla apostrofer anger att typvariabeln bara kan instantieras till en 
likhetstyp.
Listtyper är likhetstyper omm elementtypen är det!
int list är en likhetstyp, men real list är det inte.
Även likhetsjämförelser som x=[] kräver att x har likhetstyp, fastän 
det inte finns några element att jämföra.
Därför är det bättre att använda null x (eller matchning) om man 
vill testa om x är tom – null fungerar även med listor som inte har 
likhetstyp.

Definierade polymorfa funktioner
Polymorfi fungerar därför att den polymorfa funktionen (t.ex. hd) inte  
utför någon operation på värden av den obestämda typen.
Definierade funktioner blir också polymorfa om de har argument 
som de inte gör någon operation med:
fun swap(x,y) = (y,x);      Typen är 'a*'b -> 'b*'a
fun second l  = hd(tl l);   Typen är 'a list -> 'a
fun second0 l = hd(tl l)+0; Typen är int list -> int
second0 gör samma beräkning som second men är inte polymorf 
på grund av (den onödiga) additionen av 0.

Räkna antalet element i listan
length l är antalet element i listan l.
fun length l = if null l then
                 0
               else
                 1+length(tl l)

length [17,42]
––> if null[17,42] then 0 else 1+length(tl [17,42])
––> 1+length(tl [17,42])
––> 1+length [42]
––> 1+(if null [42] then 0 else 1+length(tl [42]))
––> 1+(1+length(tl [42]))
––> 1+(1+length [])
––> 1+(1+(if null [] then 0 else 1+length(tl [])))
––> 1+(1+0))
––> 2

Sätt samman två listor
append(x,y) ger en lista av elementen i x följda av elementen i y.
fun append(x,y) = if null x then
                    y
                  else
                    hd x :: append(tl x,y)
append([1,2],[3,4])
––> if null [1,2] then [3,4] else hd [1,2] ::append(tl [1,2],[3,4])
––> hd [1,2]::append(tl [1,2],[3,4])
––> 1::append([2],[3,4])
––> 1::(if null [2] then [3,4] else hd [2]::append(tl [2],[3,4])
––> 1::(hd [2]::append(tl [2],[3,4])
––> 1::(2::append([],[3,4]))
––> 1::(2::(if null [] then [3,4] else hd [] ::append(tl [],[3,4])
––> 1::(2::[3,4])
––> 1::[2,3,4]
––> [1,2,3,4]

Observera att append kopierar x, men inte y!

Mönstermatchning med listor
Listnotationen kan användas i matchning:
[1,2,3] matchar mönstret [x,y,z].
x binds till 1, y binds till 2 och z binds till 3.

:: är egentligen ingen funktion utan en (datastruktur)konstruktor.
Symbolen :: representerar en listcell – och kan också användas i 
matchning.
[1,2,3] matchar mönstret x::y.  x binds till 1, y binds till [2,3].
[1,2,3] matchar mönstret x::y::z.  x binds till 1, y binds till 2 
och z binds till [3].

Rekursion med mönstermatchning
fun length [] = 0
  | length (_::rest) = 1+length rest;

length [17,42]
––> 1+length [42]
––> 1+(1+length [])
––> 1+(1+0))
––> 2
fun append([],y) = y
  | append(first::rest, y) = first::append(rest,y)

append([1,2],[3,4])
––> 1::append([2],[3,4])
––> 1::(2::append([],[3,4]))
––> 1::(2::[3,4])
––> 1::[2,3,4]
––> [1,2,3,4]

Vänd på en lista
rev vänder bakfram på en lista.
fun rev [] = []
  | rev (first::rest) = append(rev rest,[first]);

Eftersom append kopierar sitt första argument är detta en ganska 
dålig lösning som gör en massa kopieringar i onödan.
En bättre lösning är att använda en hjälpfunktion revAux. 
revAux(l,ack) beräknar omvändningen av l följd av ack.
fun revAux([],ack) = ack
  | revAux(first::rest,ack) = 
      revAux(rest,first::ack);

fun rev l = revAux(l,[]);

revAux gör inga onödiga kopieringar.

Vänd på en lista (körexempel)
fun revAux([],ack) = ack
  | revAux(first::rest,ack) = 
      revAux(rest,first::ack)

fun rev l = revAux(l,[])

rev [1,2,3,4]
––> revAux([1,2,3,4],[])
––> revAux([2,3,4],[1])
––> revAux([3,4],[2,1])
––> revAux([4],[3,2,1])
––> revAux([],[4,3,2,1])
––> [4,3,2,1]

Hur kommer man på detta? Vi får se längre fram i kursen.

Fler inbyggda listfunktioner
length, rev och append är så vanliga att de är fördefinierade.
append heter som inbyggd funktion @ och är en infix operator:

[1,2] @ [3,4] = [1,2,3,4]

OBS! Skilj på :: (cons) och @ (append)!

:: skapar en ny listcell, medan @ sätter ihop hela listor:
 "A"   ::   ["B"]   =   ["A","B"]

            

nil

"B"    "A"

nil

"B"
["A"]   @   ["B"]   =   ["A","B"]

nil

"A"    

nil

"B"    "A"

nil

"B"
"A" @ ["B"] eller ["A"] :: ["B"] ger typfel!

Sökning i listor
Problem: Är ett bestämt värde ett element i en given lista?
Problemuppdelning: Kontrollera första elementet i listan direkt och 
resten av elementen med rekursion.
member(x,l) är true om x är ett element i l,false annars.
fun member(_,[]) = false
  | member(x,first::rest) = x = first orelse 
                             member(x,rest)
member(42,[17,42])
––> 42 = 17 orelse member(42,[42])
––> false orelse member(42,[42])
––> member(42,[42])
––> 42 = 42 orelse member(42,[])
––> true orelse member(42,[]) ––> true
Notera att member  har typen ''a * ''a list -> bool.
Den är polymorf för likhetstyper för att den jämför elementen med =.

Hanoi-programmet med listrepresentation
hanoiL(n,from,via,to) beräknar en lista som beskriver hur 
man skall flytta n brickor i Hanoi-spelet från pinne from till pinne to 
med hjälp av pinne via. Varje element i listan är ett par som anger 
att en bricka skall flyttas från pinnen som anges av första elementet 
till pinnen som anges av andra elementet.
fun hanoiL(0,_,_,_) = []
  | hanoiL(n,from,via,to) =
      hanoiL(n-1,from,to,via)
        @ (from,to) :: hanoiL(n-1,via,from,to);

hanoiL(2,"A","B","C") =
  [("A","B"),("A","C"),("B","C")]

PKD 2010/11 moment 3 Sida 55 Uppdaterad 2011-02-08



Struktur hos program som
arbetar med listor

Eftersom listor har godtycklig längd blir program som arbetar med 
listor i allmänhet rekursiva.

Oftast har rekursiva program över listor följande form:
fun f(...,[],...) = ...
  | f(...,first::rest,...) = ... f(...,rest,...) ...

Denna typ av rekursion kallas strukturell rekursion – rekursion över 
strukturen hos listan.

Strängar och teckenlistor
Strängar är en följd av tecken: "Hej, hopp"
Teckenlistor (char list):
[#"H", #"e", #"j",#" ", #"h", #"o", #"p", #"p"]
är också en följd av tecken – vad är skillnaden?
• Strängar och teckenlistor representeras (troligen) helt olika.
• Strängar kräver mindre minnesutrymme.
• Man har direkt åtkomst till delar av strängar utan att behöva "läsa 
  förbi" den del av strängen som ligger före.
• Man kan matcha listor men inte strängar.
• Rekursion över listor är mer naturlig än över strängar.
Konverteringsfunktioner:
explode: string -> char list
implode: char list -> string

Annorlunda basfall
Det är inte alltid som rekursionen går ”hela vägen” till 0, tom sträng, 
tom lista, etc...
Problem: Leta upp det största elementet i en heltalslista [x1,...,xn].
Problemuppdelning: Leta upp det största elementet  i [x2,..,xn]. Tag 
det största av detta och x1.
Hur fungerar det om det inte finns något x2?
Kan en tom lista ha ett största element?
fun largest [x] = x
  | largest (first::rest) =
      Int.max(first,largest rest);

largest [] är meningslöst och programmet fungerar inte i det 
fallet.
(Int.max beräknar det största av två heltal.)

Flera basfall
Skriv en funktion revString för att vända tecknen i en sträng! 
Problemuppdelning:
• Byt plats på första och sista tecknet
• Byt plats på tecknen emellan
Vad händer om strängen har jämnt antal tecken? Udda?
Två basfall!
fun revString "" = ""
  | revString s =
      if size s = 1 then
        s
      else
        String.substring(s,size s -1,1) ^
         revString(String.substring(s,1,size s-2))
         ^ String.substring(s,0,1)

Vektorer
• För att komma åt ett element i en lista krävs en tidsåtgång som 
  ökar när listan blir större.
• Man har ibland behov av att komma åt data i konstant tid.
• Nästan alla datorer har arbetsminne med celler ordnade 
  i följd där olika delar går att komma åt i konstant tid.
• De flesta programspråk har en datastruktur som direkt motsvarar 
   strukturen hos minnet, kallade vektorer eller arrayer.

 
 0   1   2   3    4 
18  12  ~5  12   10

En heltalsvektor med 5 element och deras index.
Till skillnad från en lista som består av ett antal celler i kedja så är 
en vektor ett objekt – att komma åt varje element går alltså lika fort.
Jämför med strängar som kan betraktas som vektorer av tecken.

Vektorer i ML
I ML skrivs vektorer på samma sätt som listor, men med tecknet # 
framför. Vektorn i exemplet skrivs  #[18, 12, ~5, 12, 10] och
har typen int vector.
Funktioner för vektorer i ML finns i biblioteket Vector. Exempel:
• Vector.fromList skapar en vektor från en lista
• Vector.length ger längden av en vektor
• Vector.sub tar fram ett element ur en vektor
• Vector.concat slår ihop vektorer
Eftersom en vektor är ett objekt så finns ingen motsvarighet till hd, 
tl, :: o.dyl. för vektorer. Man får alltså betala för åtkomst i 
konstant tid med att det är besvärligare att skapa/ändra vektorer.

Vektorfunktioner – exempel
Vector.fromList l  skapar en vektor från listan l.
   Vector.fromList [18, 12, ~5, 12, 10] =
              #[18, 12, ~5, 12, 10]

Vector.length a  ger längden av vektorn x.
   Vector.length #[18,12,~5,12,10] = 5

Vector.sub(a,i) ger elementet med index i inom vektorn a.
  Vector.sub(#[18,12,~5,12,10],2) = ~5

Vector.concat l tar en lista av vektorer l och ger en vektor som 
innehåller elementen, i tur och ordning, från vektorerna i listan l.
  Vector.concat[#[18],#[12,~5],#[],#[12,10]] =
    #[18,12,~5,12,10]

Sökning i vektorer
memberVector söker i en vektor precis som member gör i en lista.
Man kan inte enkelt kan dela upp en vektor – istället används en 
hjälpfunktion med positionsargument (som i countChar3').
memberVectorAux(x,a,i) ger true om x finns bland de i första 
elementen i vektorn a.
fun memberVectorAux(x,a,0) = false
  | memberVectorAux(x,a,i) = 
      x = Vector.sub(a,i-1) orelse 
        memberVectorAux(x,a,i-1);
fun memberVector(x,a) = 
       memberVectorAux(x,a,Vector.length a);
- memberVector(12,#[18,12,~5,12,10]);
val it = true : bool
- memberVector(7,#[18,12,~5,12,10]);
val it = false : bool

Ett större exempel – Mönstermatchning
Problem:
Man vill se om en sträng passar mot ett teckenmönster (matchar).
I detta exempel är ett mönster en följd av tecken där
• ? betyder ”godtyckligt tecken”
• * betyder ”en följd av noll eller flera godtyckliga tecken”.

Exempel:
"K*a", "K?ll? Ank?" och "Kalle*" matchar "Kalle Anka".
"K*k", "K?alle Anka" och "??" matchar inte "Kalle Anka".

Skriv en funktion som kan avgöra om ett mönster matchar strängen 
eller ej!

Problemuppdelning för matchning
Är mönstret tomt eller ej?
  a) Ja – då måste strängen också vara tom.
  b) Nej – dela upp mönstret i första tecken och resten
      Välj alternativ beroende på första tecken
      a) "?" – matcha resten av strängen med resten av mönstret.
      b) "*" – matcha stjärnan (hur?!?)
      c) Annat tecken – kolla att strängen börjar med det tecknet och
          matcha resten av strängen mot resten av mönstret.

Matcha stjärnan
Stjärnan matchar ett godtyckligt (inkl. 0) antal tecken. Ett mönster 
som börjar med en stjärna matchar alltså en sträng om resten av 
mönstret matchar strängen med början på godtycklig plats.

Man får pröva i början av strängen och sedan prova ett tecken 
framåt i taget (sökning).

T.ex. sträng: "abcd", mönster "*cd".
Matchar "abcd" mönstret "cd"? Nej.
Matchar "bcd" mönstret "cd"? Nej.
Matchar "cd" mönstret "cd"? Ja!
(Här matchade alltså stjärnan de två tecknen "ab".)

Problemuppdelning för matchning av stjärna
Har vi provat hela strängen?
  a) Ja – Mönstret matchar inte
  b) Nej. Matchar strängen vid given position resten av mönstret?
      a) Ja. Klart.
      b) Nej. Prova med nästa position.

Nu kan vi skriva kod direkt från problemuppdelningarna!
Vi får en huvudfunktion och en hjälpfunktion.

Funktionen match
match(s,pattern) ger true om s matchar pattern, false 
annars.

fun match(s,"") = s = ""
  | match(s,pattern) =
     case String.sub(pattern,0) of
     #"*" =>
      matchStar(s,
       String.substring(pattern,1,size pattern-1),
                0)
   | #"?" => s <> "" andalso
      match(String.substring(s,1,size s-1),
       String.substring(pattern,1,size pattern-1))
   | c => s <> "" andalso
      String.sub(s,0) = c andalso
      match(String.substring(s,1,size s-1),
       String.substring(pattern,1,size pattern-1))

Funktionen matchStar
matchStar(s,pattern,pos) ger true om s –– bortsett från 
pos (eller flera) tecken i början –– matchar pattern, false 
annars.

and matchStar(s,pattern,pos) =
      pos <= size s andalso
       (match(String.substring(s,pos,size s-pos),
              pattern) orelse
         matchStar(s,pattern,pos+1))

Provkörning av match....
match("abc","*bc") = true
match("bc","*bc") = true
match("abc","?bc") = true
match("bc","?bc") = false
match("Kalle Anka","K*a") = true
match("Kalle Anka","K?ll? Ank?") = true
match("Kalle Anka","Kalle*") = true 
match("Kalle Anka","K*k") = false 
match("Kalle Anka","K?alle Anka") = false 
match("Kalle Anka","??") = false 

